Pensieve header: Computing the Fibonnaci numbers.

In[1]:=
```
F[0] = F[1] = 1;
F[n_] /; n > 1 := F[n] = F[n - 1] + F[n - 2];
F /@ Range[0, 10]
```

Out[3]= {1, 1, 2, 3, 5, 8, 13, 21, 34, 55, 89}

In[4]:=
```
A = ( 0 1
      1 1 );
MatrixPower[A, 10] // MatrixForm
```

Out[5]//MatrixForm=

$$\begin{pmatrix} 34 & 55 \\ 55 & 89 \end{pmatrix}$$

In[6]:=
```
{MatrixPower[A, 50][[2, 2]], F[50]}
```

Out[6]= {20 365 011 074, 20 365 011 074}

In[7]:=
```
(B = A - λ IdentityMatrix[2]) // MatrixForm
```

Out[7]//MatrixForm=

$$\begin{pmatrix} -\lambda & 1 \\ 1 & 1-\lambda \end{pmatrix}$$

In[8]:=
```
χ = Det[B]
```

Out[8]= $-1 - \lambda + \lambda^2$

In[9]:=
```
{λ1, λ2} = λ /. Solve[χ == 0, λ]
```

Out[9]= $\left\{\frac{1}{2}\left(1-\sqrt{5}\right), \frac{1}{2}\left(1+\sqrt{5}\right)\right\}$

In[10]:=
```
(DD = DiagonalMatrix[{λ1, λ2}]) // MatrixForm
```

Out[10]//MatrixForm=

$$\begin{pmatrix} \frac{1}{2}\left(1-\sqrt{5}\right) & 0 \\ 0 & \frac{1}{2}\left(1+\sqrt{5}\right) \end{pmatrix}$$

In[11]:=
```
{v1} = NullSpace[B /. λ → λ1]
```

Out[11]= $\left\{\left\{\frac{1}{2}\left(-1-\sqrt{5}\right), 1\right\}\right\}$

In[12]:=
```
{v2} = NullSpace[B /. λ → λ2]
```

Out[12]= $\left\{\left\{\frac{1}{2}\left(-1+\sqrt{5}\right), 1\right\}\right\}$

In[13]:=
```
(CC = Transpose[{v1, v2}]) // MatrixForm
```

Out[13]//MatrixForm=

$$\begin{pmatrix} \frac{1}{2}\left(-1-\sqrt{5}\right) & \frac{1}{2}\left(-1+\sqrt{5}\right) \\ 1 & 1 \end{pmatrix}$$

In[14]:= **(CCinv = Inverse[CC]) // MatrixForm**

Out[14]//MatrixForm=

$$\begin{pmatrix} -\frac{1}{\sqrt{5}} & -\frac{1-\sqrt{5}}{2\sqrt{5}} \\ \frac{1}{\sqrt{5}} & -\frac{-1-\sqrt{5}}{2\sqrt{5}} \end{pmatrix}$$

In[15]:= **CCinv.CC // Simplify // MatrixForm**

Out[15]//MatrixForm=

$$\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}$$

In[16]:= **CC.DD.CCinv // Simplify // MatrixForm**

Out[16]//MatrixForm=

$$\begin{pmatrix} 0 & 1 \\ 1 & 1 \end{pmatrix}$$

In[17]:= **(DDn = DiagonalMatrix[{λ1^n, λ2^n}]) // MatrixForm**

Out[17]//MatrixForm=

$$\begin{pmatrix} \left(\frac{1}{2}\left(1-\sqrt{5}\right)\right)^n & 0 \\ 0 & \left(\frac{1}{2}\left(1+\sqrt{5}\right)\right)^n \end{pmatrix}$$

In[18]:= **CC.DDn.CCinv // Simplify // MatrixForm**

Out[18]//MatrixForm=

$$\begin{pmatrix} \frac{2^{-1-n}\left(\left(1-\sqrt{5}\right)^n\left(1+\sqrt{5}\right)+\left(-1+\sqrt{5}\right)\left(1+\sqrt{5}\right)^n\right)}{\sqrt{5}} & \frac{2^{-n}\left(-\left(1-\sqrt{5}\right)^n+\left(1+\sqrt{5}\right)^n\right)}{\sqrt{5}} \\ \frac{-\left(\frac{1}{2}\left(1-\sqrt{5}\right)\right)^n+\left(\frac{1}{2}\left(1+\sqrt{5}\right)\right)^n}{\sqrt{5}} & \frac{2^{-1-n}\left(-\left(1-\sqrt{5}\right)^{1+n}+\left(1+\sqrt{5}\right)^{1+n}\right)}{\sqrt{5}} \end{pmatrix}$$

In[19]:= **Formula = (CC.DDn.CCinv)[[2, 2]] // Simplify**

Out[19]=

$$\frac{2^{-1-n}\left(-\left(1-\sqrt{5}\right)^{1+n}+\left(1+\sqrt{5}\right)^{1+n}\right)}{\sqrt{5}}$$

In[20]:= **Formula /. n → 50**

Out[20]=

$$\frac{-\left(1-\sqrt{5}\right)^{51}+\left(1+\sqrt{5}\right)^{51}}{2\,251\,799\,813\,685\,248\sqrt{5}}$$

In[21]:= **Formula /. n → 50 // Expand**

Out[21]= 20 365 011 074